package iso 
{
	/**
	 * 绘制 Iso Tile 类
	 * @author StefanieKaka
	 */
	public class DrawnIsoTile extends IsoObject 
	{
		/// 构造函数
		public function DrawnIsoTile(size : Number, color : uint, height : Number = 0) 
		{
			super(size);
			m_color = color;
			m_height = height;
			draw();
		}
		
///////////////////////////////////////////////////////////////////////////////////////////////////
// 公共域
///////////////////////////////////////////////////////////////////////////////////////////////////

		/**
		 * 设置对象的颜色
		 */
		public function set color(color : uint) : void
		{
			m_color = color;
			draw();
		}
		
		public function get color() : uint
		{
			return m_color;
		}

///////////////////////////////////////////////////////////////////////////////////////////////////
// 重写函数
///////////////////////////////////////////////////////////////////////////////////////////////////	

		/**
		 * 设置对象的高度
		 */
		override public function set height(value : Number) : void
		{
			m_height = value;
			draw();
		}
		
		override public function get height() : Number
		{
			return m_height;
		}

///////////////////////////////////////////////////////////////////////////////////////////////////
// 保护域
///////////////////////////////////////////////////////////////////////////////////////////////////

		/// 绘制
		protected function draw() : void
		{
			this.graphics.clear();
			this.graphics.beginFill(m_color);
			this.graphics.lineStyle(0, 0, 0.5);
			this.graphics.moveTo(-size, 0);
			this.graphics.lineTo(0, -size * 0.5);
			this.graphics.lineTo(size, 0);
			this.graphics.lineTo(0, size * 0.5);
			this.graphics.lineTo(-size, 0);
		}

		protected var m_color	: uint = 0;
		protected var m_height	: Number = NaN;
	}

}
































///////////////////////////////////////////////////////////////////////////////////////////////////
// end file


